

USBメモリに関する実験、Q&A など

2011/08/10

MPC-2000 シリーズの USB メモリに関する動作確認実験・質問・回答などをまとめました。
(重複する内容もありますがご容赦ください)

USBメモリへの書き込み PRINT#と USB_WRITE (081202).....	1
USBメモリ読み書き時間の実測 (100107).....	5
USBメモリからプログラムを読めるか (100514).....	9
USBメモリの残容量を知りたい (110105).....	10
USBメモリファイルサイズ、読み書き時間、再読み込み (110218).....	11
USBメモリのファイル数 (110323).....	14
DIR コマンド オプション追加 (110323).....	15
DIR コマンド 所要時間の実測値 (110324).....	17
USBメモリへ PRINT# で書き込み中停止 (110429).....	18
USBメモリが動作異常 ERROR CODE 56 (110610).....	20
タイムアウト、エラーコード追加 (110726).....	21
その他のUSBメモリを使用したアプリケーションノート.....	22

USBメモリへの書き込み PRINT#と USB_WRITE (081202)

MPCのUSBメモリ書き込みは512バイトでバッファリングされます。
そのため APPEND→PRINT# で書き込む場合、CLOSE 前に電源が落ちるとバッファリングされていたデータは消失します。
(CLOSE 前に電源が切れても既にUSBメモリに書き込まれたデータは無くなりません)
USB_WRITE は APPEND→書き込み→CLOSE を一連で行うのでリスクが低くなります。

DIR

ドライブ A のボリュームラベルは TOSHIBA です

2008/12/02 10:20	512 AA.CSV	←512+α バイト書いたが
2008/06/25 18:17	171 MARU.P2K	CLOSE せず電源をオフした
2008/06/25 17:59	98 SHIKAKU.P2K	ので+α 分が消えた
2008/08/22 16:39	119 ZIGZAG.P2K	
2008/06/24 17:22	19 ZUPDOWN.P2K	
2008/11/20 10:41	393089 2000-002.PDF	
	7 個のファイル	394008 バイト
	0 個のディレクトリ	

A:>

下記はアプリケーションノート[an2k-002]のプログラムです。 ※がUSB関係の部分です。

PRINT#の場合USBメモリのアクセスランプの点灯は間欠的ですがUSB_WRITEはその都度の書き込むのでランプ点灯もそれに同期します。

参照 アプリケーションノート <http://deparonline.jp/mpc2000/ref/headline/appendix/pdf/an2k-002.pdf>

◆PRG1) PRINT#で書き込み

CUNET 4 4 8	/* CUnet 初期化
MEWNET 38400 2	/* タッチパネル通信
S_MBK &H20 8	/* タッチパネルページ切り替え
CNFG# 5 RS485 "9600b7pes1NONE"	/* MRS-MCOM ch5 RS-485 モード
OFF 70000	/* START SW オフ
FOR I=100 TO 131	/* タッチパネル使用エリア初期化
S_MBK 0 I	
NEXT I	
S_MBK 0 200	/* Write Count
S_MBK 0 300	/* タッチパネル Msg "Push START"
*START	
WAIT SW(70000)=1	/* タッチパネル START SW (オルタネート)
S_MBK 1 300	/* タッチパネル Msg "Logging"
FILENAME\$="test.csv"	/* ファイル名

```

USB_DEL_FILENAME$ /* USB メモリ 既存ファイル削除 ※
APPEND_FILENAME$ /* USB メモリ 追記モードでオープン ※

FORMAT "0000/00/00" /* 文字列書式
DT$=HEX$(DATE(0))
S_MBK DT$ 124 10 /* タッチパネル 年月日表示
ST$="時間, 湿度 1, 湿度 2, 湿度 2, "+DT$+"¥n" /* scv の表題 ※
PRINT# USB ST$ /* USB メモリ 書き込み ※
DELAY=75 /* read 間隔 必要に応じて加減
WRITE_COUNT=0

DO
  CH$="01" /* 温湿度計 ch1
  GOSUB *GET_DATA
  RES1$=RES$
  S_MBK RES1$ 100 9 /* タッチパネル 温度, 湿度表示
  OUT t 2032~Int /* CUnet に温度書き込み
  OUT h 2034~Int /* CUnet に湿度書き込み
  TIME DELAY /* この温湿度計には必要みたい

  CH$="02" /* 温湿度計 ch2
  GOSUB *GET_DATA
  RES2$=RES$
  S_MBK RES2$ 108 9
  OUT t 2036~Int
  OUT h 2038~Int

  FORMAT "00:00:00"
  TM$=HEX$(TIME(0))
  S_MBK TM$ 116 8 /* タッチパネル 時間表示
  ST$=TM$+", "+RES1$+", "+RES2$+"¥n"
  PRINT ST$
  IF (TIME(0)&&HF)%2==0 THEN /* 2 秒間隔
    INC WRITE_COUNT /* カウンタインクリメント
    S_MBK WRITE_COUNT 200~Wrd /* タッチパネル カウンタ値表示
    PRINT "WRITE TO USB-MEM" WRITE_COUNT
    PRINT# USB ST$ /* USB メモリ 書き込み ※
    WAIT (TIME(0)&&HF)%2!=0
  END_IF

  TIME DELAY /* この温湿度計には必要みたい
  IF SW(70000)==0 THEN /* タッチパネル STOP SW 押下なら ※
    CLOSE USB /* USB メモリクローズ ※
    S_MBK 2 300 /* タッチパネル Msg "USB Close"
    GOTO *START
  END_IF

  HH=TIME(0)/&H10000 /* MPC の日時を CUnet に書き込む
  MM=TIME(0)/&H100
  MM=MM&&HFF
  SS=TIME(0)&&HFF
  OUT HH 2040
  OUT MM 2041
  OUT SS 2042

LOOP

*GET_DATA
PRINT# 5 CHR$(5) CH$ CHR$(2) "RPV01" CHR$(3) "¥r¥n" /* 温湿度計にコマンド送信
INPUT# 5 a$ /* 温湿度計データ受信
dummy=VAL(a$)
FOR I=1 to 9
  dummy=VAL(0)
NEXT I
t=VAL(10) /* 温度値取得
dummy=VAL(0) /* 読み飛ばし
h=VAL(10) /* 湿度値取得
FORMAT "00.0" /* 書式=整数 2 桁、小数 1 桁
RES$=STR$(t) /* 温度文字列
FORMAT "00.0"
RES$=RES$+", "+STR$(h) /* 温度文字列, 湿度文字列
RETURN

```

◆PRG2) USB_WRITE で書き込み

```

CUNET 4 4 8 /* CUnet 初期化
MEWNET 38400 2 /* タッチパネル通信
S_MBK &H20 8 /* タッチパネルページ切り替え
CNFG# 5 RS485 "9600b7pes1NONE" /* MRS-MCOM ch5 RS-485 モード
OFF 70000 /* START SW オフ

*START
FOR I=100 TO 131 /* タッチパネル使用エリア初期化
  S_MBK 0 I
NEXT I
S_MBK 0 200 /* Write Count
S_MBK 0 300 /* タッチパネル Msg "Push START"
WAIT SW(70000)=1 /* タッチパネル START SW (オルタネート)
S_MBK 1 300 /* タッチパネル Msg "Logging"
FILE$="test.csv" /* ファイル名。FILE$は予約変数
USB_DEL FILE$ /* USB メモリ 既存ファイル削除 ※
FORMAT "0000/00/00" /* 文字列書式
DT$=HEX$(DATE(0)) /* タッチパネル 年月日表示
S_MBK DT$ 124 10 /* 時間, 温度 1, 湿度 1, 温度 2, 湿度 2, "+DT$+"¥n" /* scv の表題
ST$="時間, 温度 1, 湿度 1, 温度 2, 湿度 2, "+DT$+"¥n" /* USB 書き込み。ファイル名は FILE$ ※
USB_WRITE ST$ /* read 間隔 必要に応じて加減
DELAY=75
WRITE_COUNT=0

DO
  CH$="01" /* 温湿度計 ch1
  GOSUB *GET_DATA
  RES1$=RES$
  S_MBK RES1$ 100 9 /* タッチパネル 温度, 湿度表示
  OUT t 2032~Int /* CUnet に温度書き込み
  OUT h 2034~Int /* CUnet に湿度書き込み
  TIME DELAY /* この温湿度計には必要みたい

  CH$="02" /* 温湿度計 ch2
  GOSUB *GET_DATA
  RES2$=RES$
  S_MBK RES2$ 108 9
  OUT t 2036~Int
  OUT h 2038~Int

  FORMAT "00:00:00"
  TM$=HEX$(TIME(0))
  S_MBK TM$ 116 8 /* タッチパネル 時間表示
  ST$=TM$+", "+RES1$+", "+RES2$+"¥n"
  PRINT ST$
  IF (TIME(0)&&HF)%2==0 THEN /* 2 秒間隔
    INC WRITE_COUNT /* カウンタインクリメント
    S_MBK WRITE_COUNT 200~Wrd /* タッチパネル カウンタ値表示
    PRINT "WRITE TO USB-MEM" WRITE_COUNT
    USB_WRITE ST$ /* USB 書き込み。ファイル名は FILE$ ※
    WAIT (TIME(0)&&HF)%2!=0
  END_IF

  TIME DELAY /* この温湿度計には必要みたい
  IF SW(70000)==0 THEN /* タッチパネル STOP SW 押下なら
    GOTO *START
  END_IF

  HH=TIME(0)/&H10000 /* MPC の日時を CUnet に書き込む
  MM=TIME(0)/&H100
  MM=MM&HFF
  SS=TIME(0)&&HFF
  OUT HH 2040
  OUT MM 2041
  OUT SS 2042

LOOP


```

```

*GET_DATA
PRINT# 5 CHR$(5) CH$ CHR$(2) "RPV01" CHR$(3) "¥r¥n" /* 温湿度計にコマンド送信
INPUT# 5 a$ /* 温湿度計データ受信
dummy=VAL(a$)
FOR I=1 to 9
  dummy=VAL(0)
NEXT I
t=VAL(10) /* 温度値取得
dummy=VAL(0) /* 読み飛ばし
h=VAL(10) /* 湿度値取得
FORMAT "00.0" /* 書式=整数2桁、小数1桁
RES$=STR$(t) /* 温度文字列
FORMAT "00.0"
RES$=RES$+", "+STR$(h) /* 温度文字列, 湿度文字列
RETURN

```

■PRG2 で 2 秒間隔 12000 個のログをグラフにしたものです。



※注意

USBメモリは個々の動作スピードバラツキがあります(結構違います)。
 USB_WRITE コマンドはオープン→ライト→クローズを連続で行います。
 連続してコマンドを実行する場合には十分間隔を空けてください。
 参考:手持ちの遅いUSBメモリで TIME 100 以上 でした。

USB メモリ読み書き時間の実測 (100107)

MPC-1000 で USB メモリの読み書き時間の実測を行いました。(MPC-1000 は MRS-MCOM より高速です)

■使用機器

MPC-1000
バージョン:MPC-1000 (SH7030) BL/I 1.12_15 2009/12/28
タッチパネル
パナソニック GT11
USB メモリ
1: Transcend 1GB
2: SanDisk 2GB
3: A-DATA 2GB
4: SONY 2GB
5: TOSHIBA 2GB
6: SONY 8GB

左から USB メモリ 1~5
(6は写っていませんが、4と同タイプです)



■プログラム

/* 点データ、MBK データ 各 1000 個を 書き込み→読み込みします。
/* USB メモリの R/W とは関係ありませんが、別タスクでタッチパネルから時間データ取得
/* と表示を行っています

```
MEWNET 38400 2
```

```
QUIT 10  
FORK 10 *CLOCK_DISP /* タッチパネル  
ON_USB  
TIME 500  
TCNT=0
```

```
DO  
PR "¥nTOTAL=" TCNT  
TCNT=TCNT+1  
  
FOR I=1 TO 1000 /* MPC 内部のデータ作成  
SETP I I I+1 I+2 I+3  
S_MBK I I+5000  
NEXT I  
PR "DATA CREATE"
```

```
GOSUB *PNT_SAVE  
IF SW(197)==0 THEN  
BREAK  
END_IF
```

```
GOSUB *MBK_SAVE  
IF SW(197)==0 THEN  
BREAK  
END_IF
```

```
LOOP
```

```
END
```

```
*PNT_SAVE  
PR "-- PNT DATA --"
```

```
SYSCLK=0  
USB_DEL "PNTSAVE.P2K" /* USB メモリの既存ファイル削除  
PR "DELETE" SYSCLK
```

```

SYSCLK=0
USB_PSAVE P(1) 1000 "PNTSAVE.P2K" /* USB メモリへ点データ書き込み
TIME 100 /* ※1 NEED →下記留意点
PR "PNTSAVE" SYSCLK

NEWP
PR "DATA CLEAR"

SYSCLK=0
USB_PLOAD "PNTSAVE.P2K" /* USB メモリから点データ読み込み
PR "PLOAD" SYSCLK

FOR I=1 TO 1000 /* データチェック
  IF X(I) <> I THEN
    PR "X ERROR"
  END
END_IF
  IF Y(I) <> (I+1) THEN
    PR "Y ERROR"
  END
END_IF
  IF U(I) <> (I+2) THEN
    PR "U ERROR"
  END
END_IF
  IF Z(I) <> (I+3) THEN
    PR "Z ERROR"
  END
END_IF

NEXT I
PR "CHECK OK"

RETURN

*MBK_SAVE

PR "--- MBK DATA ---"

SYSCLK=0
USB_DEL "MBKSAVE.MBK" /* USB メモリの既存ファイル削除
PR "DELETE" SYSCLK

SYSCLK=0
USB_PSAVE MBK(5001) 1000 "MBKSAVE.MBK" /* USB メモリへ MBK データ書き込み
TIME 100 /* ※1 NEED →下記留意点
PR "MBKSAVE" SYSCLK

FILL MBK(5001) 1000 0
PR "DATA CLEAR"

SYSCLK=0
USB_PLOAD "MBKSAVE.MBK" /* USB メモリから MBK データ読み込み
PR "MBKLOAD" SYSCLK

FOR I=1 TO 1000 /* データチェック
  PR I+5000 MBK(I+5000)
  IF MBK(I+5000) <> I THEN
    PR "MBK ERROR"
  END
END_IF
NEXT I
PR "CHECK OK"

RETURN

*CLOCK_DISP

```

```

TIME 500
TODAY=&H20000000+MBK(2)*&H100
TODAY=TODAY+(MBK(1)/&H100)
NOW=(MBK(1)&&HFF)*&H10000+MBK(0)
SET_RTC TODAY
SET_RTC NOW
S_MBK &H85 8          /* 時計表示画面
DO
  FORMAT "0000/00/00"
  DD$=HEX$(DATE(0))
  S_MBK DD$ 124 10     /* 年月日表示
  FORMAT "00:00:00"
  TT$=HEX$(TIME(0))
  S_MBK TT$ 116 8     /* 時分秒表示
  TIME 100
LOOP

```

■USBメモリ別実測結果

USBメモリ機種によって所要時間(特に書き込み)が異なります。
(なかでも3の書き込みが遅い。理由はわかりません)

1: Transcend 1GB

```

-- PNT DATA --
DELETE 37
PNTSAVE 4213          /* 点データ書き込み 4.2sec
DATA CLEAR
PLOAD 594            /* 点データ読み込み 0.6sec
CHECK OK
-- MBK DATA --
DELETE 37
MBKSAVE 2568         /* MBKデータ書き込み 2.6sec
DATA CLEAR
MBKLOAD 391         /* MBKデータ読み込み 0.4sec
CHECK OK

```

2: SanDisk 2GB

```

-- PNT DATA --
DELETE 59
PNTSAVE 4585         /* 点データ書き込み 4.6sec
DATA CLEAR
PLOAD 626           /* 点データ読み込み 0.6sec
CHECK OK
-- MBK DATA --
DELETE 60
MBKSAVE 3152        /* MBKデータ書き込み 3.2sec
DATA CLEAR
MBKLOAD 412        /* MBKデータ読み込み 0.4sec
CHECK OK

```

3: A-DATA 2GB

```

-- PNT DATA --
DELETE 41
PNTSAVE 12968       /* 点データ書き込み 13.0sec
DATA CLEAR
PLOAD 763           /* 点データ読み込み 0.8sec
CHECK OK
-- MBK DATA --
DELETE 41
MBKSAVE 8286        /* MBKデータ書き込み 8.3sec
DATA CLEAR
MBKLOAD 568        /* MBKデータ読み込み 0.6sec
CHECK OK

```

4: SONY 2GB

```

-- PNT DATA --
DELETE 41

```

```

PNTSAVE 4160          /* 点データ書き込み  4.2sec
DATA CLEAR
PLOAD 580             /* 点データ読み込み  0.6sec
CHECK OK
-- MBK DATA --
DELETE 40
MBKSAVE 2723         /* MBK データ書き込み  2.7sec
DATA CLEAR
MBKLOAD 392         /* MBK データ読み込み  0.4sec
CHECK OK

```

5: TOSHIBA 2GB

```

-- PNT DATA --
DELETE 46
PNTSAVE 4179         /* 点データ書き込み  4.2sec
DATA CLEAR
PLOAD 587           /* 点データ読み込み  0.6sec
CHECK OK
-- MBK DATA --
DELETE 42
MBKSAVE 3248        /* MBK データ書き込み  3.2sec
DATA CLEAR
MBKLOAD 389        /* MBK データ読み込み  0.4sec
CHECK OK

```

6: SONY 8GB

```

-- PNT DATA --
DELETE 54
PNTSAVE 8153        /* 点データ書き込み  8.2sec
DATA CLEAR
PLOAD 602           /* 点データ読み込み  0.6sec
CHECK OK
-- MBK DATA --
DELETE 194
MBKSAVE 5482        /* MBK データ書き込み  5.4sec
DATA CLEAR
MBKLOAD 392        /* MBK データ読み込み  0.4sec
CHECK OK

```

■留意点

USB メモリ機種によって、書き込み後の待機時間が異なるようです。
(バッファデバイス書き込み時間の違いなどによるものと思われます)
それを考慮し書き込み後に※1のタイマーを入れています。
この実験のように連続でR/Wを行う場合、※1のTIMEが無いとエラーになります。
タイマー値は使用するUSBメモリや実行手順で加減してください。
今回の実験に用いたUSBメモリではTIME 50でも良好でしたが、安全のため長めの設定を推奨します。
下記はMBKデータの読み込みを失敗していますが、USBメモリ内のデータは正常でした。

エラー時の実行結果

```

TOTAL= 141
DATA CREATE
-- PNT DATA --
DELETE 37
PNTSAVE 4010
DATA CLEAR
PLOAD 584
CHECK OK
-- MBK DATA --
DELETE 37
MBKSAVE 2507
DATA CLEAR
MBKLOAD 30          /* 時間短すぎ=読み込み異常
MBK ERROR          /* データ比較でもエラー

```

■参考 : USBメモリ関係コマンド

#OFF_USB /* USB deactivate

#VER

MPC-1000(SH7030) BL/I 1.12_15 2009/12/28

All Rights reserved. ACCEL Corp. T32

[* and / performed before + or -]

#ON_USB /* USB activate

#VER

MPC-1000(SH7030) BL/I 1.12_15 2009/12/28

All Rights reserved. ACCEL Corp. T32

[* and / performed before + or -]

+The USB Activated on TASK_29+ /* ON_USB するとこの表示が追加される

#DIR /* ファイル一覧(ルートのみ)

```
2010/01/06 17:09 26483 PNTSAVE.P2K
2010/01/06 17:08 16893 MBKSAVE.MBK
2009/11/12 14:36 741 DOREMI.P2K
2009/11/12 14:47 1560 TOTORO.P2K
2009/11/12 14:35 702 CHARA.P2K
2009/11/06 16:39 2036 PONYO.P2K
2009/11/11 09:37 <DIR> DEMO_M~1
6 個のファイル 48415 バイト
1 個のディレクトリ
```

A:>

#TYPE "PNTSAVE.P2K" /* ファイル内容表示。ファイル名は""でくる

```
SETP 1 1 2 3 4
SETP 2 2 3 4 5
SETP 3 3 4 5 6
SETP 4 4 5 6 7
SETP 5 5 6 7 8
SETP 6 6 7 8 9
SETP 7 7 8 9 10
SETP 8 8 9 10 11
SETP 9 9 10 11 12
SETP 10 10 11 12 13
SETP 11 11 12 13 14
SETP 12 12 13 14 15
SETP 13 13 14 15 16
SETP 14 14 15 16 17
SETP 15 15 16 17 18
SETP 16 16 17 18 19
SETP 17 17 18 19 20
SETP 18 18 19 20 21
SETP 19 19 20 21 22
SETP 20 20 21 22 23
SETP 21 21 22 23 24
SETP 22 22 23 24 25
```

/* "Q"キーで終了、その他のキーで次表示

#

USBメモリからプログラムを読めるか (100514)

USBメモリからUSB_LOADコマンドをダイレクト実行すればプログラム(.F2K)が読み込めます。USB_LOADコマンドをプログラムに入れても動作しますが、プログラムの入れ替え・実行(オーバーレイみたいな)はできません。

例えばFREEZEと併用して、作業プログラムの入れ替えることはできません。データ変更は点データまたはMBKデータを使用します(USB_PLOADコマンド)。

保守や機種切り換えのためにプログラム自体を入れ替えるのはDSWの切り換え方式です。(マニュアル MRS-MCOM 参照)

USBメモリの残容量を知りたい (110105)

USBメモリの残量を知るにはDIRコマンドとUSB関数を使用します。
(このサンプルは2GのUSBメモリを使用しました)

DIRコマンドをダイレクト実行すると現状を表示します。

```
#DIR
2010/02/22 18:25          137 SQUARE.P2K
2010/02/23 15:20          160 INFINI.P2K
2010/02/23 15:21          307 OVAL_GCW.P2K
2010/02/23 15:21          256 OVAL_CW.P2K
      4 個のファイル          860 バイト
      0 個のディレクトリ
A:>
```

DIR 1000 というようにパラメータを付けて実行するとMBKエリアにデータが格納されます。

```
MBK(1000) にファイル数
MBK(1001) にトータルファイル数
MBK(1002) にトータルディレクトリ数
MBK(1003) にUSB使用量(Mbytes、ただし ルートディレクトリの分のみ)
```

たとえば、上記のUSBメモリの場合は次のようになります。

```
#DIR 1000
#PR MBK(1000) MBK(1001) MBK(1002) MBK(1003)
  4 4 0 1
```

また、USBメモリ総量はUSB関数で取得できます。

```
#PR USB(1, USB)
 1967
```

つまり残量は
総量-使用量 = USB(1, USB)-MBK(1003) で求められます。

```
・プログラム例
10 DIR 1000
20 total_mem=USB(1, USB)
30 PRINT "TOTAL(MBytes)" total_mem
40 available_mem=total_mem-MBK(1003)
50 PRINT "AVAILABLE(MBytes)" available_mem
#RUN
```

```
TOTAL(MBytes) 1967
AVAILABLE(MBytes) 1966
#
```

【注意】
MPCで扱えるのはルートディレクトリのファイルみです。
使用量もルートディレクトリの分だけです。
MBKエリアはタッチパネル等と重複しないように設定してください。

USB メモリ ファイルサイズ、読み書き時間、再読み込み (110218)

【 Q 】 USB メモリ内のファイル容量の制約は?

【 A 】 ファイルサイズに特に制限を設けていませんが、USB メモリは動作が遅く、メーカーや容量によりバラツキがあり、書き込み回数にも限界があります。
データをまとめて巨大なファイルで扱うと何らかの理由で壊れたりしたとき困ります。
実際に使用する USB メモリで想定される最大サイズの動作確認をお勧めします。

【 Q 】 テキストデータの書き込み/読み込み時間とファイルサイズの関係は?

【 A 】 下記の実行結果のように、ファイルサイズが大きくなっても追記に要する時間はさほどかわりませんが、INPUT#読み込みはサイズが大きいほど遅くなります。

【 Q 】 データを再度読み出すには?

【 A 】 サンプルプログラムは key 3 押下で何回でもファイル先頭から読み出せます。

■MPC プログラム

```
/*  
* USB メモリ 書き込み/読み込みテスト  
*/
```

```
FILENAME$="test.txt"  
WRITE_COUNT=0
```

```
DO  
  PRINT "1:WRITE with DEL"  
  PRINT "2:WRITE without DEL"  
  PRINT "3:READ"
```

```
  INPUT CHR_C|1 key$  
  PRINT "--->KEY=" key$  
  SELECT_CASE key$  
    CASE "1" : GOSUB *WRITE  
    CASE "2" : GOSUB *WRITE  
    CASE "3" : GOSUB *READ  
    CASE_ELSE : PRINT "input again"  
  END_SELECT  
LOOP
```

```
*WRITE  
IF key$=="1" THEN  
  PRINT "DELETE"  
  USB_DEL FILENAME$  
  WRITE_COUNT=0  
END_IF
```

```
APPEND FILENAME$  
COUNT=0  
WRITE_COUNT=WRITE_COUNT+1  
PRINT "WRITE" WRITE_COUNT "回目"  
SYSCLK=0  
DO  
  COUNT=COUNT+1  
  IF COUNT>10000 THEN  
    BREAK  
  END_IF  
  FORMAT ""  
  ST$=STR$(COUNT)  
  FORMAT "00"  
  ST$=ST$+" "+HEX$(SYSCLK&&HFF)+"¥n"  
  PRINT# USB ST$  
LOOP  
CLOSE USB  
PRINT "所要時間" SYSCLK  
DIR /* 確認表示  
RETURN
```

```
*READ  
OPEN USB FILENAME$  
PRINT "READ"
```

```

SYSCLK=0
DO
  IF LOF(USB)==0 THEN
    BREAK
  END IF
  TIME 1 /* delay need
  INPUT# USB a$
  a=VAL(a$)
  IF a%1000==0 THEN
    PRINT a$
  END IF
LOOP
CLOSE USB
PRINT "所要時間" SYSCLK

RETURN

```

■MPC プログラムの実行結果

MPC-2100L MPC-2100L (SH7030) BL/I 1.12_59 2010/12/28
 USB メモリは MRS-MCOM でアクセス

実験 USB メモリ
 SONY 2GB
 SanDisk 2GB (下記結果はこれ)

所要時間単位 mSec

RUN

```

1:WRITE with DEL
2:WRITE without DEL
3:READ
--->KEY= 1
DELETE ←ファイル削除
WRITE 1 回目 ←書込み1回目
所要時間 8035 ←所要時間
2001/01/08 13:38 88894 TEST.TXT ←作成ファイル
                88894 バイト
                1 個のファイル
                0 個のディレクトリ

A:>
1:WRITE with DEL
2:WRITE without DEL
3:READ
--->KEY= 3
READ ←読み込み
1000 47 ←1000単位
2000 4C
3000 F4
4000 FE
5000 86
6000 1A
7000 10
8000 B2
9000 C0
10000 42
所要時間 20010 ←所要時間
1:WRITE with DEL
2:WRITE without DEL
3:READ
--->KEY= 2
WRITE 2 回目 ←書込み2回目 追記
所要時間 10489
2001/01/08 13:38 177788 TEST.TXT
                177788 バイト
                1 個のファイル
                0 個のディレクトリ

```

```

A:>
1:WRITE with DEL
2:WRITE without DEL
3:READ

```

--->KEY= 2
WRITE 3 回目 ←書込み3回目 追記
所要時間 10115
2001/01/08 13:38 266682 TEST.TXT
1 個のファイル 266682 バイト
0 個のディレクトリ

A:>
1:WRITE with DEL
2:WRITE without DEL
3:READ
--->KEY= 3
READ ←読み込み

1000 47
2000 4C
3000 F4
4000 FE
5000 86
6000 1A
7000 10
8000 B2
9000 C0
10000 42
1000 57
2000 80
3000 43
4000 66
5000 2A
6000 6F
7000 15
8000 DD
9000 0F
10000 D8
1000 D7
2000 81
3000 CE
4000 71
5000 B5
6000 85
7000 2B
8000 72
9000 1A
10000 62

所要時間 60010 ←所要時間
1:WRITE with DEL
2:WRITE without DEL
3:READ
*0 [70] ←Ctrl+A 停止

#

USBメモリのファイル数 (110323)

これは MRS-MCOM で実験した結果です。
用いたファイルのサイズは小さいものです。
また、USBメモリによっては違いがあるかもしれませんので、使用する場合は事前に動作確認をしてください。
動作保証はできません。

使用機器

MPC-2100L (SH7030) BL/I 1.12_62
MRS-MCOM Ver 20090514
SanDisk 2GB USBメモリ

USBメモリをWindows PCで
FAT16 (Windows7ではFAT)、
アロケーションユニットサイズ 32キロバイト
でフォーマットした場合は最高512ファイルです。
513個目を書き込もうとするとAPPENDで戻ってきません。

FAT32
アロケーションユニットサイズ 4096バイト
でフォーマットした場合は沢山入ります。

下記のプログラムで1024個までWRITE/READを5回実行 (APPENDなのでファイルサイズは1回ごと大きくなる) しましたが良好でした。

```
*WRITE
FORMAT "00000000"
FOR i=1 TO 64
  FILENAME$=STR$(i)+".TXT"
  CONTENT$=STR$(i)+":"+TIME$(0)
  APPEND FILENAME$
  PRINT# USB CONTENT$
  CLOSE
  TIME 1000
  PR "Write FileName " FILENAME$
/*  TYPE FILENAME$
/*  TIME 500
NEXT
```

```
*CHECK
FORMAT "00000000"
FOR i=1 TO 64
  FILENAME$=STR$(i)+".TXT"
  PR "Check FileName " FILENAME$
  OPEN USB FILENAME$
  INPUT# USB a$
  CLOSE
  PR a$
  IF i<>VAL(a$) THEN
    PR "Error " i
  END
  END IF
  TIME 1000

NEXT
END
```

DIR コマンド オプション追加 (110323)

DIR コマンドを

DIR mbkst

と実行すると MBK エリアに下記のように USB メモリの情報が入ります。

```
MBK (mbkst)   ファイル数
MBK (mbkst+1) トータルファイル数
MBK (mbkst+2) トータルディレクトリ数
MBK (mbkst+3) 使用量
MBK (mbkst+4) ~ ファイル名 (1 ファイル名は 12 キャラ=6 ワード つまり MBK () 6 個使用)
```

ファイル数が多いと MBK エリアを大量消費してしまいます。
そこで DIR コマンドに使用範囲の限定オプションがつけました。
従来の使い方でも最大 50 個になりました。

◆新バージョン

MPC-2100L (SH7030) BL/I 1.12_64 2011/03/22

DIR [USB] mbkst

範囲指定無しで 50 個分の範囲

MBK (mbkst) ~ MBK (mbkst+304) の間におさめる。
それより多い場合は最後のファイルに上書き

DIR [USB] mbkst mbken

MBK (mbkst) ~ MBK (mbken) の範囲におさめる
それより多い場合は最後のファイルに上書き

下のプログラムは 64 個のファイルが入っている USB メモリで実行しました。
USB メモリは MRS-MCOM (Ver 20090514) でアクセスしました。

```
#dir                               /* FTMW 画面上ファイル一覧表示
2011/03/23 09:18                   17 00000001.TXT
2011/03/23 09:18                   17 00000002.TXT
(中略)
2011/03/23 09:22                   17 00000063.TXT
2011/03/23 09:22                   17 00000064.TXT
                                64 個のファイル
                                0 個のディレクトリ
A:>
```

◆プログラム 1

```
10   mbkst=1000
20   FOR i=mbkst TO mbkst+387
30     S_MBK 0 i
40   NEXT
50   DIR USB mbkst /* DIR のファイル名エリア範囲指定無し
60   PRINT "MBK (" mbkst ") " "Files " MBK (mbkst)
70   PRINT "MBK (" mbkst+1 ") " "Total Files" MBK (mbkst+1)
80   PRINT "MBK (" mbkst+2 ") " "Total Dir " MBK (mbkst+2)
90   PRINT "MBK (" mbkst+3 ") " "Using " MBK (mbkst+3)
100  FOR i=mbkst+4 TO mbkst+384 STEP 6
110  PRINT "MBK (" i ") " "FileName" MBK $(i, 12)
120  NEXT
130  END
```

◆実行結果 1

```
MBK ( 1000 ) Files 64
MBK ( 1001 ) Total Files 64
MBK ( 1002 ) Total Dir 0
MBK ( 1003 ) Using 1
MBK ( 1004 ) FileName 00000001.TXT
MBK ( 1010 ) FileName 00000002.TXT
MBK ( 1016 ) FileName 00000003.TXT
(中略)
MBK ( 1286 ) FileName 00000048.TXT
MBK ( 1292 ) FileName 00000049.TXT
MBK ( 1298 ) FileName 00000064.TXT /* Max50、後ろのファイル名は上書き
```

```
MBK( 1304 ) FileName
MBK( 1310 ) FileName
(以下略)
```

◆プログラム 2

```
10      mbkst=1000
20      FOR   i=mbkst TO mbkst+387
30          S_MBK  0 i
40      NEXT
50      DIR   USB mbkst mbkst+33      /* DIR のファイル名エリア範囲指定有り
60      PRINT "MBK(" mbkst ") " "Files " MBK(mbkst)
70      PRINT "MBK(" mbkst+1 ") " "Total Files" MBK(mbkst+1)
80      PRINT "MBK(" mbkst+2 ") " "Total Dir " MBK(mbkst+2)
90      PRINT "MBK(" mbkst+3 ") " "Using " MBK(mbkst+3)
100     FOR   i=mbkst+4 TO mbkst+384 STEP 6
110     PRINT "MBK(" i ") " "FileName" MBK$(i,12)
120     NEXT
130     END
#
```

◆実行結果 2

```
MBK( 1000 ) Files 64
MBK( 1001 ) Total Files 64
MBK( 1002 ) Total Dir 0
MBK( 1003 ) Using 1
MBK( 1004 ) FileName 00000001.TXT
MBK( 1010 ) FileName 00000002.TXT
MBK( 1016 ) FileName 00000003.TXT
MBK( 1022 ) FileName 00000004.TXT
MBK( 1028 ) FileName 00000064.TXT      /* 後ろのファイル名は上書き
MBK( 1034 ) FileName
MBK( 1040 ) FileName
(以下略)
```

◆プログラム 3

```
10      mbkst=1000
20      FOR   i=mbkst TO mbkst+387
30          S_MBK  0 i
40      NEXT
50      DIR   USB mbkst mbkst+3      /* 範囲最小 0~ でも...
60      PRINT "MBK(" mbkst ") " "Files " MBK(mbkst)
70      PRINT "MBK(" mbkst+1 ") " "Total Files" MBK(mbkst+1)
80      PRINT "MBK(" mbkst+2 ") " "Total Dir " MBK(mbkst+2)
90      PRINT "MBK(" mbkst+3 ") " "Using " MBK(mbkst+3)
100     FOR   i=mbkst+4 TO mbkst+384 STEP 6
110     PRINT "MBK(" i ") " "FileName" MBK$(i,12)
120     NEXT
130     END
```

◆実行結果 3

```
MBK( 1000 ) Files 64
MBK( 1001 ) Total Files 64
MBK( 1002 ) Total Dir 0
MBK( 1003 ) Using 1
MBK( 1004 ) FileName 00000064.TXT      /* ファイル名 1 個は入る
MBK( 1010 ) FileName
MBK( 1016 ) FileName
(以下略)
```


DIR コマンド 所要時間の実測値 (110324)

USB メモリの DIR コマンド所要時間の実測値です。
USB メモリ : 2GB、FAT32 フォーマット
これらの時間は USB メモリによって異なります。

◆MPC プログラム

```
10      ON_USB
20      mbkst=1000
30      FOR   i=mbkst TO mbkst+15
40          S_MBK  0 i
50      NEXT
60      SYSCLK=0
70      DIR   USB mbkst mbkst+15
80      PRINT "spent time " SYSCLK
90      PRINT "MBK(" mbkst ") " "Files " MBK(mbkst)
100     PRINT "MBK(" mbkst+1 ") " "Total Files" MBK(mbkst+1)
110     PRINT "MBK(" mbkst+2 ") " "Total Dir " MBK(mbkst+2)
120     PRINT "MBK(" mbkst+3 ") " "Using " MBK(mbkst+3)
130     FOR   i=mbkst+4 TO mbkst+10 STEP 6
140         PRINT "MBK(" i ") " "FileName" MBK$(i, 12)
150     NEXT
160     END
```

◆実行結果

* MPC-2100L + MRS-MCOM
(MPC-2100L (SH7030) BL/I 1.12_64 2011/03/22 、 MRS-MCOM 20090514)

```
ファイル数 64
spent time 1009          /* 約 1.0sec
MBK( 1000 ) Files 64
MBK( 1001 ) Total Files 64
MBK( 1002 ) Total Dir 0
MBK( 1003 ) Using 1
MBK( 1004 ) FileName 00000001.TXT
MBK( 1010 ) FileName 00000064.TXT
```

```
ファイル数 512
spent time 2586          /* 約 2.6sec
MBK( 1000 ) Files 512
MBK( 1001 ) Total Files 512
MBK( 1002 ) Total Dir 0
MBK( 1003 ) Using 1
MBK( 1004 ) FileName 00000001.TXT
MBK( 1010 ) FileName 00000512.TXT
```

```
ファイル数 1024
spent time 5195          /* 約 5.2sec
MBK( 1000 ) Files 1024
MBK( 1001 ) Total Files 1024
MBK( 1002 ) Total Dir 0
MBK( 1003 ) Using 1
MBK( 1004 ) FileName 00000001.TXT
MBK( 1010 ) FileName 00001024.TXT
```

* MPC-1000
(MPC-1000 (SH7030) BL/I 1.12_64 2011/03/22)

```
ファイル数 64
spent time 956           /* 約 1.0sec
MBK( 1000 ) Files 64
MBK( 1001 ) Total Files 64
MBK( 1002 ) Total Dir 0
MBK( 1003 ) Using 1
MBK( 1004 ) FileName 00000001.TXT
MBK( 1010 ) FileName 00000064.TXT
```

ファイル数 512

```
spent time 1538 /* 約1.5sec
MBK( 1000 ) Files 512
MBK( 1001 ) Total Files 512
MBK( 1002 ) Total Dir 0
MBK( 1003 ) Using 1
MBK( 1004 ) FileName 00000001.TXT
MBK( 1010 ) FileName 00000512.TXT
```

```
ファイル数 1024
spent time 2207 /* 約2.2sec
MBK( 1000 ) Files 1024
MBK( 1001 ) Total Files 1024
MBK( 1002 ) Total Dir 0
MBK( 1003 ) Using 1
MBK( 1004 ) FileName 00000001.TXT
MBK( 1010 ) FileName 00001024.TXT
```

USBメモリへ PRINT# で書き込み中停止 (110429)

■ユーザーからのご報告

PRINT# で繰り返し書き込んでいるとたまに戻ってこなくなることがある。

- ・ボード：MRS-MCOM
- ・USBメモリ仕様：不明
- ・MPCプログラム詳細不明
- ・1m延長ケーブル使用

■ベンチテスト

下記のMPCプログラムで繰り返し書き込みを行いましたがご指摘の不具合は再現できませんでした。

■確認・対策案

- ・USB_WRITE コマンドで書き込んでみたらどうでしょうか(サブルーチン *WRITE2)。USB_WRITE は open → apped write → close をします。PRINT#より遅いです。
- ・延長ケーブルにも電源・信号の減衰、ノイズ混入の懸念があります。USBメモリを直接ボードに差す、短いケーブルにする、引き回しを変える、ケーブル品質を確認する等をしてみてください。(“たまに”というのが怪しい)。フェライトコア付きのケーブルはどうでしょう?(弊社で試したことはありませんが)
- ・USBメモリ自体にも品質にばらつきがあります。品質の確かなメーカーのものを使用してください

■ベンチテストMPCプログラム

- ・使用したUSBメモリ
- | | |
|---------------|-------|
| SanDisk 2GB | FAT32 |
| imation 512MB | FAT32 |
| ADATA 2GB | FAT |

```
FILENAME$="WR_TEST.TXT"
```

```
DO
```

```
PRINT "1:WRITE1"
PRINT "2:WRITE2"
PRINT "3:READ"
```

```
INPUT CHR_C|1 key$
```

```
PRINT "---->KEY=" key$
```

```
SELECT_CASE key$
```

```
CASE "1" : GOSUB *WRITE1
```

```
CASE "2" : GOSUB *WRITE2
```

```
CASE "3" : GOSUB *READ
```

```
CASE_ELSE : PRINT "input again"
```

```
END_SELECT
```

```
LOOP
```

```
*WRITE1
```

```
USB_DEL USB FILENAME$ /* delete a file
```

```
PR FILE$ "Deleted"
```

```
APPEND FILENAME$ /* create a file and append open
```

```
FORMAT ""
```

```
FOR i=0 TO 10000
```

```
PR i
```

```
CONTENT$=STR$(i)+": "+HEX$(TIME(0))+"$r$rn"
```

```

CONTENT$=CONTENT$+"123456789012345678901234567890123456789012345678901234567890¥r¥n"
PRINT# USB CONTENT$      /* write

NEXT
CLOSE USB

RETURN

*WRITE2

FILE$=FILENAME$          /* FILE$ is a particular filename for the USB_WRITE command

USB_DEL USB FILE$        /* delete a file
PR FILE$ "Deleted"
/*APPEND FILE$           /* create a file and append open
/*CLOSE USB              /* close port

FORMAT ""
FOR i=0 TO 10000
  PR i
  CONTENT$=STR$(i)+": "+HEX$(TIME(0))+¥r¥n"
  CONTENT$=CONTENT$+"123456789012345678901234567890123456789012345678901234567890¥r¥n"

  USB_WRITE CONTENT$     /* open -> write -> close

NEXT

RETURN

*READ

OPEN USB FILENAME$
DO
  IF LOF(USB)==0 THEN
    BREAK
  END IF
  TIME 1                 /* delay need
  INPUT# USB a$
  PRINT a$
LOOP
CLOSE USB

RETURN

```

USBメモリが動作異常 ERROR CODE 56 (110610)

■ERROR 56 の発生事例と対策です。

(A) SanDisk のエラー その(1)

現象：電源投入直後の ※A(DIR)で ERROR 56 発生。

対策：SanDiskのリセットタイミングの問題と思われます。*ERROR_PROC の RST_USB コマンドで復帰し、その後は正常に動作します。USBメモリをパソコンでフォーマットしても改善されません。

(B) SanDisk のエラー その(2)

現象：USB_WRITE コマンドで書き込む場合、既存ファイルが無ければ新規ファイルを作成するはずなのに、それが出来ずに※B1(USB_WRITE)で ERROR 56 が発生しました。※B2(APPEND と CLOSE)を有効にすると書き込めるようになりました。

対策：USBメモリをパソコンでフォーマットし直したら※B2が無くても正常に動作するようになりました。

(C) Sony のエラー

現象：※C1(TIME)が無い場合、その次の ※C2(DIR)で ERROR 56 が発生しました。*ERROR_PROC を数回繰り返すと復帰しましたが、その後も ERROR 56 になりました。

対策：USBメモリをパソコンでフォーマットし直したら※C1が無くても正常に動作するようになりました。

■備考

- ・上記以外のUSBメモリでは不具合は発生しませんでした。
- ・これらのUSBメモリは実験や動作確認に用いて、何回も書き込み・消去を繰り返し、パソコンと使いまわしたりもしました。それにより、ファイルが断片化されてアクセス遅延などが発生したのかもしれませんが(推測)。
- ・どのUSBメモリもファイル数は10個以下でしたが、フォーマット後は動作速度が速くなった様に思われます。
- ・USBメモリの動作は遅いです。USBメモリコマンドを連続して行う場合は余裕のあるTIMEを入れてください。

■ベンチテスト MPC プログラム

- ・ボード
MPC-2100L(SH7030) BL/I 1.12_75 2011/06/09
MRS-CMOM Ver20090514
- ・使用したUSBメモリ
SanDisk 2GB FAT32
imation 512MB FAT32
ADATA 2GB FAT
Sony 2GB FAT32

```
MEWNET 38400
S_MBK &H30 0 /* one of a page in a touch panel display for testing
```

```
ON_ERROR *ERROR_PROC
```

```
DO
S_MBK "PUSH START" 100 10 /* touch panel display
OFF 70000
WAIT SW(70000)==1 /* start switch (alternate type)
c=0
```

```
GOSUB *CLR_TOUCH_PANEL
```

```
DIR 1000 /* ※A
```

```
PRX DATE(0) TIME(0)
IF (DATE(0)&&HFFFFFF00)<>&H20110600 THEN /* check RTC
S_MBK "DATE ERR" 100 10
S_MBK DATE$(0) 110 10
S_MBK TIME$(0) 120 10
END
END_IF
```

```
FILE$=DATE$(0)+".csv" /* create file name
```

```
/* APPEND FILE$ /* ※B2
/* CLOSE USB /* ※B2
```

```
DO
c=c+1
S_MBK DATE$(0) 110 10
```

```
S_MBK TIME$(0) 120 10
S_MBK c 220
```

```
DATE1$="123456789012345678901234567890"
DATE2$="ABCDEFGHIJKLMNPOQRSTUVWXYZ"
INDEX$="4"
```

```
USB_DATA$=DATE$(0)+", "+STR$(TIME(VOID))+", "+INDEX$+", "+DATE1$+", "+DATE2$+CHR$(13)+CHR$(10)
PR USB_DATA$
USB_WRITE USB_DATA$ /* ※B1 write and close. file name is FILE$
```

```
TIME 100 /* ※C1
DIR 1000 /* ※C2
f=MBK(1000)
PR f "files"
PR MBK(1003) "Mbytes"
PR MBK$(1004, 12)
f=f-1
PR MBK$(1004+(f*6), 12)
```

```
IF SW(70000)==0 THEN
BREAK
END_IF
```

```
LOOP
```

```
LOOP
```

```
END
```

```
*ERROR_PROC
```

```
err_code=err_>>24 /* get an error code
MBK(200~Lng)=err_code /* display an error code
MBK(210~Lng)=err_&&H00FFFFFF /* display a stopping step number
PR err_code MBK(210~Lng)
S_MBK "ERROR STOP" 100 10
OFF 70000
WAIT SW(70000)==1 /* restart button
GOSUB *CLR_TOUCH_PANEL
```

```
RST_USB /* reset the USB memory
TIME 1000
RESUME /* retry of the error step
```

```
END
```

```
*CLR_TOUCH_PANEL
```

```
S_MBK 0 200~Lng /* clear display
S_MBK 0 210~Lng
S_MBK 0 220
S_MBK " " 100 10
S_MBK " " 110 10
S_MBK " " 120 10
```

```
RETURN
```

タイムアウト、エラーコード 追加 (110726)

BL/I 1.12_81 2011/07/07 で APPEND、PRINT# 等 USB コマンドにタイムアウトを追加しました。
それに伴い、エラーコード 68 「USB メモリ応答なし」も追加されました。

<社内実験>

数万回に 1 回 APPEND でエラーが発生する 2G USB メモリにおいて次のエラー処理で復帰することを確認しました(前記「USB メモリが動作異常」参照)。

```
TIME 1000
RST_USB /* reset the USB memory
TIME 1000
RESUME /* retry of the error step
```

ただし、このメモリは 2 回のリトライで復帰、その後は正常動作します。

その他の USB メモリを使用したアプリケーションノート

3 軸直行型ロボットで小型卓上機 (MPC-1000 デモ機)

<http://deparonline.jp/mpc2000/ref/headline/appendix/pdf/an2k-018.pdf>

ステップモーターオルゴール (MPC-1000 デモ機)

<http://deparonline.jp/mpc2000/ref/headline/appendix/pdf/an2k-015.pdf>

--- End Of Document ---